


APX ATARI® PROGRAM EXCHANGE



William D. Volk

FORTH TURTLE GRAPHICS PLUS

Turtle graphics for use with EXTENDED fig-FORTH

Diskette: 24K (APX-20157)

User-Written Software for ATARI Home Computers

FORTH TURTLE GRAPHICS PLUS

by

William D. Volk

Program and Manual Contents © 1982 William D. Volk

Copyright notice. On receipt of this computer program and associated documentation (the software), the author grants you a nonexclusive license to execute the enclosed software. This software is copyrighted. You are prohibited from reproducing, translating, or distributing this software in any unauthorized manner.

TRADEMARKS OF ATARI

The following are trademarks of Atari, Inc.

ATARI
ATARI 400 Home Computer
ATARI 800 Home Computer
ATARI 410 Program Recorder
ATARI 810 Disk Drive
ATARI 820 40-Column Printer
ATARI 822 Thermal Printer
ATARI 825 80-Column Printer
ATARI 830 Acoustic Modem
ATARI 850 Interface Module

Distributed by

The ATARI Program Exchange
P. O. Box 427
155 Moffett Park Drive, B-1
Sunnyvale, CA 94086

To request an APX Software Catalog, write to the address above, or call toll-free:

800/538-1862 (outside California)
800/672-1850 (within California)

Or call our Sales number, 408/745-5535.

TABLE OF CONTENTS

INTRODUCTION	__ 1
Overview	__ 1
Required accessories	__ 1
Optional accessories	__ 1
Contacting the author	__ 1
GETTING STARTED	__ 2
Loading instructions	__ 2
Diagram conventions	__ 2
THE COORDINATE SYSTEM	__ 3
SETTING A WINDOW	__ 4
TURTLE FUNCTIONS	__ 5
MOVE	__ 5
TURN	__ 5
TURNT0	__ 5
PEN	__ 5
PENUP	__ 5
PENDN	__ 5
GOXY	__ 5
DRAWTO	__ 5
CLEAR	__ 5
FRAMEIT	__ 5
CENTER	__ 5
BOX	__ 5
CUBE	__ 6
CIRCLE	__ 6
*SIN, *COS, *TAN	__ 6
DUPMOVE	__ 6
GRx	__ 6
SCREEN DESCRIPTIONS	__ 7
SAMPLE APPLICATIONS	__ 8
Typical LOGO function	__ 8
Using FORTH TURTLE GRAPHICS PLUS WITH OTHER FORTHS AND OTHER COMPUTERS	__ 8
Sample FORTH turtle commands	__ 8
Sample graphics mode 8 screen dumps	__ 8
SUGGESTED REFERENCES	__ 11
SCREEN LISTINGS	__ 12



INTRODUCTION

OVERVIEW

FORTH TURTLE GRAPHICS PLUS is a package of FORTH "words" (functions) that adds the standard turtle graphics of languages like ATARI PILOT and LOGO to EXTENDED fig-FORTH, available through APX. Implemented in integer math, the package also supports a normalized coordinate system, the ability to create windows with clipped lines on any part of the screen display without having to modify a program, the twelve graphics modes available with the new GTIA chip, and the TRIG functions *SIN, *COS, and *TAN for use in FORTH programs. Also included is a graphics mode 8 screen dump utility for Epson printers with GRAFTRAX. Yet with all these features, FORTH TURTLE GRAPHICS PLUS compiles to only 2K of memory!

REQUIRED ACCESSORIES

24K RAM (more for some graphics modes)
Extended fig-Forth (APX-20029)
ATARI 810 Disk Drive

OPTIONAL ACCESSORIES

EPSON MX-80 or 100 Printer with GRAFTRAX

CONTACTING THE AUTHOR

This package is an evolving entity. If you have suggestions or improvements, please write to the author at:

Box 2053 Babcock House
Durham, NH 03824

GETTING STARTED

LOADING INSTRUCTIONS

FORTH TURTLE GRAPHICS PLUS must be compiled by EXTENDED fig-Forth before you can use it. You should be familiar with the EXTENDED Fig-Forth instructions before continuing.

1. Remove any cartridge from the cartridge slot of your computer.
2. Turn on your disk drive.
3. When the BUSY light goes out, open the disk drive door and insert the EXTENDED fig-Forth diskette in your disk drive.
4. Turn on your computer and TV set.
5. After the Forth "ok" prompt appears, load the assembler and graphics screens with the commands

```
39 LOAD
50 LOAD
```

6. Remove the EXTENDED fig-FORTH diskette from the disk drive and insert the FORTH TURTLE GRAPHICS PLUS diskette.

7. Type

```
1 LOAD
```

which loads in screens 1 - 7 .

8. If your ATARI Computer has a GTIA chip installed, you may want to load screen 8 also.

DIAGRAM CONVENTIONS

Each function description includes a FORTH stack diagram, showing the effect of a function on the stack. The left-hand side of the diagram shows the state of the parameter stack before executing the function and the right-hand side of the diagram shows the state of the parameter stack after executing the function. For example, the addition function diagram looks like this:

```
+ ( n m --- n+m )
```

The diagram indicates the stack contains the operands "n" and "m" before executing the addition function and the single operand "n+m" after executing the function.

THE COORDINATE SYSTEM

While you're using FORTH TURTLE GRAPHICS PLUS, the coordinate system is fixed and independent of the ATARI Computer's graphic coordinates. It is centered on 0,0 and extends to +/-1000 along both the X and Y axes. Figure 1 depicts the range along both axes.




Figure 1 Range along X and Y axes

The two screen modes best used with FORTH TURTLE GRAPHICS PLUS are BASIC modes 7 and 8. These modes are accessed by the functions GR8 and GR7, which set the window to the entire graphic display. The GTIA modes are on screen 8, which includes the functions GR9 and GR11. To use other modes, use the GR. function (mode ---), as described in the EXTENDED fig-Forth documentation, and set an appropriate window.

SETTING A WINDOW

WINDOW (LeftX TopY RightX BottomY ---)

Sets a window on the screen. The coordinates are given in ATARI BASIC screen coordinates. For example, a GR@ command executes a [0 0 319 159 WINDOW].

Warning. Once a window is set, all turtle commands take place in the +/-1000 coordinate system. A distorted window introduces distortion in the picture.

TURTLE FUNCTIONS

MOVE (x ---)

Move in the current direction x units.

TURN (x ---)

Turn x degrees to the right.

TURNTO (x ---)

Turn to a direction of x degrees.

PEN (x ---)

Select pen color x and place the pen down. The color corresponds to the COLOR command in ATARI BASIC.

PENUP (---)

Lift the pen up.

PENDN (---)

Place the pen down.

GOXY (x y ---)

Go to the coordinate x, y with the pen up.

DRAWTO (x y ---)

Go to the coordinate x, y with the pen down.

CLEAR (n ---)

Fill the window with color n.

FRAMEIT (n ---)

Draw a box around the window in color n.

CENTER (---)

The same as a 0 0 GOXY command.

BOX (n ---)

Draw a box of dimension n.

CUBE (n ---)

Draw a cube of dimension n.

CIRCLE (n ---)

Draw a circle with n as the scaling factor.

*SIN, *COS, and *TAN (n ang --- n*SIN[ang]), etc.

Multiply a number by a trig function. These functions multiply an integer with the trig function at a given angle (in degrees). This is accomplished with a 91-element table of integers (0-32767) and the */ function in FORTH.

DUPMOVE (n --- n)

Move in the current direction n units and preserve the stack.

GRx (---)

Defined for modes 7, 8, 9, and 11. Sets the window to the entire screen. For example, the window for GR8 is set to 0,0,319,159.

SCREEN DESCRIPTIONS

Screen 1

Contains the code to create the trig table used by *SIN, *COS, and *TAN. This table is located by the variable TRIGTAB.

Screen 2

The recursion function RECURSE is defined on line 0. This allows a FORTH word to call itself. Recursion is used by the function ANGSET, which converts any angle to an angle within 0-359 degrees. The trig functions are defined on this screen. The trig functions *SIN, *COS, and *TAN are implemented by doing a look-up on TRIGTAB.

Screen 3

The variables used by FORTH TURTLE GRAPHICS PLUS are defined here. Some functions are also defined. The coordinate size, +/- 1000, is set on line 15.

Screen 4

Some of the clipping functions, the variables used by the clipping function, and the transformation to the window are defined here.

Screen 5

This screen contains the bulk of the clipping function, which clips lines to the +/- 1000 window. It is based on the FORTRAN clipping program in A Practical Introduction to Computer Graphics. This results in a larger FORTH function than is normally desirable.

Screen 6

The remaining turtle functions are defined here. Some examples of turtle functions (BOX, CUBE, and CIRCLE) are also defined.

Screen 7

The CLEAR and CENTER functions are defined here.

Screen 8

The GTIA extensions are defined, along with the demonstrations GSQUIRAL, BOXES, SPIRAL, DEMO1, and DEMO2. All demonstrations except DEMO2 have a stack diagram (n ---), specifying the number of iterations. (DEMO2 does not affect the stack.)

Screen 9

This screen defines GRDUMP (---), which sends a graphics mode 8 screen dump to an Epson printer with GRAFTRAX. You must execute a PON command before dumping. The data is echoed on the bottom of the screen; the data may include the bell character.

SAMPLE APPLICATIONS

TYPICAL LOGO FUNCTION

The typical LOGO function SQUIRAL can be defined as follows:

```
: SQUIRAL 1 DO I 10 * MOVE 123 TURN LOOP ; ( n --- )
```

A typical use of this package would be a graphical adventure game. Objects could be defined as a collection of turtle commands. The power of the package is that the objects can be translated or rotated simply by executing a GOXY and TURNTO command. The windowing lets you change graphics modes or screen format easily. An excellent example would be to provide a GTIA version of a program.

The package can also be used to generate pie charts and other graphical reports. It might be nice to allow the user to specify the location and size of the window.

Multiple windows are possible. By storing the parameters for the WINDOW command in an array in memory, you could switch from one window to another. If the windows overlap, the current window will have precedence.

USING FORTH TURTLE GRAPHICS PLUS WITH OTHER FORTHS AND OTHER COMPUTERS

FORTH TURTLE GRAPHICS PLUS is very transportable to other versions of FORTH and to other computers. Only five functions supported by the package are unique to EXTENDED fig-FORTH: DRAW, GR., XGR, PLOT, and COLOR. And the package contains only three functions--DLINE, GRx, and PEN--that are dependent on functions defined in EXTENDED fig-FORTH. For example, DLINE, which draws lines, uses PLOT and DRAW. Therefore, to transport graphics, you need only define these functions for the other system or replace them in the code with corresponding functions for other systems. Transportability of the graphics in a program gives you the major advantage of being able to move software to new environments easily.

SAMPLE FORTH TURTLE COMMANDS

Try these commands. Type them in sequence, pressing RETURN at the end of each line:

```
: INIT GR8 90 0 210 100 WINDOW 1 DUP PEN FRAMEIT ;  
  
INIT 100 @CIRCLE  
INIT 500 @BOX  
INIT 400 @CUBE  
INIT 1 CLEAR 0 CLEAR
```

SAMPLE GRAPHICS MODE 8 SCREEN DUMPS


Using an Epson printer with GRAFTRAX, you can obtain a screen dump of the sample commands. First, type

9 LOAD

to load in the screen containing the graphics mode 8 screen dump function. Then to dump the circle command, type

```
GR8 INIT 100 CIRCLE P0N GRDUMP P0FF
```


The figure looks like this:



To dump the BOX command, type

```
GR8 INIT 500 BOX P0N GRDUMP P0FF
```


The figure looks like this:



To dump the CUBE command, type


```
GR8 INIT 400 CUBE P0N GRDUMP P0FF
```

The figure looks like this:



Here's a sample dump of the demonstration DEMO1 from screen 8, obtained with the command:

```
GRS PON 40 DEMO GRDUMP POFF
```



SUGGESTED REFERENCES

1. MIND-STORMS: Children, Computers, and Powerful Ideas, Seymour Papert.

A good introduction to turtle graphics and procedural programming.

2. Starting FORTH, Leo Brodie, Prentice-Hall, 1981.

The best introduction to FORTH programming.

3. A Practical Introduction to Computer Graphics, Ian O. Angell.

Good discussion of clipping functions. Contains FORTRAN listings of programs.

4. Fundamentals of Interactive Computer Graphics, J.D. Foley & A. Van Dam.

Excellent graphics text. Good explanation of windowing and its uses.

Additional documentation on fig-FORTH is available from the FORTH Interest Group, P.O. Box 1105, San Carlos, CA 94070.



SCR # 1

```
0 0 VARIABLE TRIGTAB 571 , 1143 , 1714 , 2285 , 2855 ,
1 3425 , 3993 , 4560 , 5125 , 5689 , 6252 , 6812 ,
2 7370 , 7927 , 8480 , 9031 , 9580 , 10125 , 10667 ,
3 11206 , 11742 , 12274 , 12803 , 13327 , 13847 ,
4 14364 , 14875 , 15383 , 15885 , 16383 , 16876 ,
5 17363 , 17846 , 18323 , 18794 , 19259 , 19719 ,
6 20173 , 20620 , 21062 , 21497 , 21925 , 22347 ,
7 22761 , 23169 , 23570 , 23964 , 24350 , 24729 ,
8 25100 , 25464 , 25820 , 26168 , 26509 , 26841 ,
9 27165 , 27480 , 27787 , 28086 , 28377 , 28658 ,
10 28931 , 29195 , 29450 , 29696 , 29934 , 30162 ,
11 30381 , 30590 , 30790 , 30981 , 31163 , 31335 ,
12 31497 , 31650 , 31793 , 31927 , 32050 , 32164 ,
13 32269 , 32363 , 32448 , 32522 , 32587 , 32642 ,
14 32687 , 32722 , 32747 , 32762 , 32767 ,
15 -->
```

SCR # 2

```
0 : RECURSE LATEST PFA CFA , ; IMMEDIATE ( RECURSION )
1 32767 CONSTANT MAXINT ( LARGEST INTERGER )
2 : GETRIG 2 * TRIGTAB + @ ; ( RETURNS TABLE ELEMENT N )
3 : ANGSET DUP 359 > IF 360 - RECURSE ELSE DUP
4 0 < IF 360 + RECURSE ENDIF ENDIF ; ( GET ANGLE 0-359 )
5 : SGNSIN ANGSET 180 > IF MINUS ENDIF ; ( GIVE SIGN TO VALUE )
6 : SGNCOS 90 + SGNSIN ;
7 : INDEXSIN DUP 180 > IF 180 - ENDIF DUP 90 > IF 180 SWAP -
8 ENDIF ; ( GET THE INDEX OF THE SIN FUNCTION )
9 : INDEXCOS INDEXSIN 90 SWAP - ; ( GET COS OFFSET )
10 : GETSIN ANGSET DUP INDEXSIN GETRIG SWAP SGNSIN ;
11 : GETCOS ANGSET DUP INDEXCOS GETRIG SWAP SGNCOS ; ( VALUES )
12 : *SIN GETSIN MAXINT */ ; ( N ANG --- N*SIN )
13 : *COS GETCOS MAXINT */ ; ( N ANG --- N*COS )
14 : *TAN DUP GETSIN SWAP GETCOS */ ; ( N ANG --- N*TAN )
15 -->
```

SCR # 3

```
0 ( TURTLE GRAPHICS WILLIAM VOLK 3/29/82 )
1 0 VARIABLE TURTANG ( ANGLE OF MOTION )
2 : TURN TURTANG @ SWAP - ANGSET TURTANG ! ; ( TURN TURTLE )
3 : TURNT0 ANGSET TURTANG ! ;
4 0 VARIABLE TURTCOL ( COLOR OF PEN )
5 0 VARIABLE LCORX 0 VARIABLE LCORY 0 VARIABLE RCORX
6 0 VARIABLE RCORY 0 VARIABLE X1 0 VARIABLE Y1 0 VARIABLE X2
7 0 VARIABLE Y2 0 VARIABLE PENDOWN ( 0 PEN IS UP, 1 DOWN )
8 : PEN TURTCOL ! 1 PENDOWN ! ; ( SET COLOR PEN IS DOWN )
9 : PENDOWN? PENDOWN @ ; ( LOGICAL TEST OF PEN )
10 : PENUP 0 PENDOWN ! ; ( LIFT THE PEN UP )
11 : WINDOW RCORY ! RCORX ! LCORY ! LCORX ! ; ( SET SCREEN )
12 : GR7 XGR 7 GR. 0 0 159 79 WINDOW ;
13 : GR8 XGR 8 GR. 0 0 319 159 WINDOW ;
14 : PENDN 1 PENDOWN ! ; ( PLACE THE PEN DOWN )
15 1000 CONSTANT PICSIZE -->
```

```

SCR # 4
0 ( MORE TURTLE GRAPHICS )
1 0 VARIABLE X1D 0 VARIABLE Y1D 0 VARIABLE X2D
2 0 VARIABLE Y2D : XFORM PICSIZE SWAP - RCORY @ LCORY @ - 2000 */
3 LCORY @ + SWAP PICSIZE + RCORX @ LCORX @ - 2000 */ LCORX @ +
4 SWAP ; : DLINE TURTCOL @ X1D @ Y1D @ XFORM PLOT TURTCOL @ X2D @
5 Y2D @ XFORM DRAW ;
6 : MODE OVER ABS < IF DUP ABS / ELSE DROP 0 ENDIF ;
7 ( X Y -- INX INY ) ( CLIPPING PARAMETERS 1 0 -1 )
8 0 VARIABLE IX1 0 VARIABLE IY1 0 VARIABLE IX2 0 VARIABLE IY2
9 ( CLIPPING XN YN -- IXN IYN )
10 : CMODE PICSIZE ROT OVER MODE ROT ROT MODE SWAP ;
11 : NMODE X2D @ Y2D @ CMODE IX2 ! IY2 ! ;
12 : OMODE X1D @ Y1D @ CMODE IX1 ! IY1 ! ;
13 ( THE CLIPPING FUNCTION )
14 : ICLIP X1 @ X1D ! Y1 @ Y1D ! X2 @ X2D ! Y2 @ Y2D ! ;
15 ( SET CLIPPED CO-ORDS TO LINE CO-ORDS ) -->

```

```

SCR # 5
0 : CLIP ICLIP NMODE OMODE ( GET CLIP PARAMS )
1 IX1 @ IX2 @ * 1 = 0= ( IF IX1*IX2 <> 1 )
2 IY1 @ IY2 @ * 1 = 0= ( AND IY1*IY2 <> 1 )
3 AND IF IX1 @ 0= 0= IF ( THEN IF IX1 <> 0 )
4 PICSIZE IX1 @ * DUP ( XX=DX*IX1 )
5 X1D @ - Y2D @ Y1D @ - SWAP X2D @ X1D @ - */ Y1D @ + Y1D !
6 X1D ! OMODE ENDIF IY1 @ 0= 0= IF ( IF IY1 <> 0 THEN )
7 PICSIZE IY1 @ * DUP ( YY=DY*IY1 )
8 Y1D @ - X2D @ X1D @ - SWAP Y2D @ Y1D @ - */ X1D @ + X1D !
9 Y1D ! ENDIF IX2 @ 0= 0= IF ( IF IX2 <> 0 )
10 PICSIZE IX2 @ * DUP ( XX=DX*IX2 )
11 X1D @ - Y2D @ Y1D @ - SWAP X2D @ X1D @ - */ Y1D @ + Y2D !
12 X2D ! NMODE ENDIF IY2 @ 0= 0= IF ( IF IY2 <> 0 )
13 PICSIZE IY2 @ * DUP ( YY=DY*IY2 )
14 Y1D @ - X2D @ X1D @ - SWAP Y2D @ Y1D @ - */ X1D @ + X2D !
15 Y2D ! ENDIF X1D @ X2D @ = Y1D @ Y2D @ = AND 0= IF DLINE -->

```

```

SCR # 6
0 ENDIF ENDIF ;
1 : MOVE DUP TURTANG @ *COS X1 @ + X2 !
2 TURTANG @ *SIN Y1 @ + Y2 ! PENDOWN? IF CLIP ENDIF
3 Y2 @ Y1 ! X2 @ X1 ! ;
4 : DUPMOVE DUP MOVE ; ( MAINTAINS THE MOVEMENT ON STACK )
5 : BOX 4 0 DO DUPMOVE 90 TURN LOOP DROP ;
6 : CIRCLE 18 0 DO DUPMOVE 20 TURN LOOP DROP ;
7 : CUBE DUPMOVE 90 TURN DUPMOVE 45 TURN DUPMOVE
8 45 TURN DUPMOVE 90 TURN DUPMOVE 45 TURN
9 DUPMOVE 135 TURN DUPMOVE -90 TURN DUPMOVE 180 TURN
10 PENUP DUPMOVE PENDN -45 TURN MOVE -135 TURN ;
11 : GOXY Y1 ! X1 ! ; ( MOVE CURSOR )
12 : DRAWTO Y2 ! X2 ! CLIP ; ( ABSOLUTE DRAW )
13 : FRAMEIT DUP LCORX @ LCORY @ PLOT DUP RCORX @ LCORY @ DRAW
14 DUP RCORX @ RCORY @ DRAW DUP LCORX @ RCORY @ DRAW
15 LCORX @ LCORY @ DRAW ; ( FRAME WINDOW COLOR X ) -->

```

```

SCR # 7
0 ( TURTLE GRAPHIC FUNCTIONS )
1 : CLEAR RCORY @ 1 + LCORY @ DO ( CLEAR WINDOW )
2 DUP LCORX @ I PLOT DUP RCORX @ I DRAW
3 LOOP DROP ;
4 : CENTER 0 0 GOXY ;
5
6
7
8
9
10
11
12
13
14
15

```

```

SCR # 8
0 ( GITA DEMOS )
1 : GR9 XGR 9 GR. 0 0 79 191 WINDOW CENTER ;
2 : GR11 XGR 11 GR. 0 0 79 191 WINDOW CENTER ;
3 : GSQUIRAL 1 DO I PEN I 10 * MOVE 123 TURN LOOP ;
4 : BOXES 1 DO I PEN I 10 * BOX LOOP ;
5 : SPIRAL 1 DO I PEN I MOVE 30 TURN LOOP ;
6 : DEMO1 1 DO I PEN I 10 * BOX 10 TURN LOOP ;
7 : DEMO2 12 1 DO I 1 - CLEAR I PEN 700 CUBE LOOP ;
8
9
10
11
12
13
14
15

```

```

SCR # 9
0 ( PRINT DUMP GR8 )
1 : ROWDUMP -1 160 DO DUP 88 @ + I 40 * + @ EMIT -1 +LOOP DROP ;
2 : SETDUMP ." " 27 EMIT 75 EMIT 160 EMIT ;
3 : GRDUMP 27 EMIT 65 EMIT 8 EMIT CR 40 0 DO SETDUMP I ROWDUMP
4 CR LOOP ;
5 : DEMO CENTER 1 DO I 20 * BOX 10 TURN LOOP ; ( GRAPHICS )
6
7
8
9
10
11
12
13
14
15

```



LIMITED WARRANTY ON MEDIA AND HARDWARE ACCESSORIES.

We, Atari, Inc., guarantee to you, the original retail purchaser, that the medium on which the APX program is recorded and any hardware accessories sold by APX are free from defects for thirty days from the date of purchase. Any applicable implied warranties, including warranties of merchantability and fitness for a particular purpose, are also limited to thirty days from the date of purchase. Some states don't allow limitations on a warranty's period, so this limitation might not apply to you. If you discover such a defect within the thirty-day period, call APX for a Return Authorization Number, and then return the product along with proof of purchase date to APX. We will repair or replace the product at our option.

You void this warranty if the APX product: (1) has been misused or shows signs of excessive wear; (2) has been damaged by use with non-ATARI products; or (3) has been serviced or modified by anyone other than an Authorized ATARI Service Center. Incidental and consequential damages are not covered by this warranty or by any implied warranty. Some states don't allow exclusion of incidental or consequential damages, so this exclusion might not apply to you.

DISCLAIMER OF WARRANTY AND LIABILITY ON COMPUTER PROGRAMS.

Most APX programs have been written by people not employed by Atari, Inc. The programs we select for APX offer something of value that we want to make available to ATARI Home Computer owners. To offer these programs to the widest number of people economically, we don't put APX products through rigorous testing. Therefore, APX products are sold "as is", and we do not guarantee them in any way. In particular, we make no warranty, express or implied, including warranties of merchantability and fitness for a particular purpose. We are not liable for any losses or damages of any kind that result from use of an APX product.



ATARI PROGRAM EXCHANGE

REVIEW FORM

We're interested in your experiences with APX programs and documentation, both favorable and unfavorable. Many software authors are willing and eager to improve their programs if they know what users want. And, of course, we want to know about any bugs that slipped by us, so that the software author can fix them. We also want to know whether our documentation is meeting your needs. You are our best source for suggesting improvements! Please help us by taking a moment to fill in this review sheet. Fold the sheet in thirds and seal it so that the address on the bottom of the back becomes the envelope front. Thank you for helping us!

1. Name and APX number of program _____

2. If you have problems using the program, please describe them here.

3. What do you especially like about this program?

4. What do you think the program's weaknesses are?

5. How can the catalog description be more accurate and/or comprehensive?

6. On a scale of 1 to 10, 1 being "poor" and 10 being "excellent", please rate the following aspects of this program?

- _____ Easy to use
- _____ User-oriented (e.g., menus, prompts, clear language)
- _____ Enjoyable
- _____ Self-instructive
- _____ Useful (non-game software)
- _____ Imaginative graphics and sound

7. Describe any technical errors you found in the user instructions (please give page numbers).

8. What did you especially like about the user instructions?

9. What revisions or additions would improve these instructions?

10. On a scale of 1 to 10, 1 representing "poor" and 10 representing "excellent", how would you rate the user instructions and why?

11. Other comments about the software or user instructions:

STAMP

ATARI Program Exchange
Attn: Publications Dept.
P.O. Box 50047
60 E. Plumeria Drive
San Jose, CA 95150

[seal here]